IMPLEMENTATION OF THE IRS DYNAMIC REDUCTION METHOD IN MSC/NASTRAN

Christopher C. Flanigan
Director, Aerospace Projects
SDRC Engineering Services Division, Inc.
San Diego, California

1990 MSC/NASTRAN World Users Conference
Los Angeles, California
March 26-30, 1990

IMPLEMENTATION OF THE IRS DYNAMIC REDUCTION METHOD IN MSC/NASTRAN

Christopher C. Flanigan
SDRC Engineering Services Division, Inc.
San Diego, California

Abstract

The Improved Reduced System (IRS) dynamic reduction method is a recent extension of the Guyan reduction method. The IRS method includes mass effects in the development of the matrix reduction transformation matrix. This improvement can significantly increase the matrix reduction accuracy for certain types of models. The IRS method can be especially useful for test/analysis correlation to accurately reduce the FEM matrices to DOF corresponding to accelerometer locations. The IRS dynamic reduction method is implemented in MSC/NASTRAN using a rigid format alter to a normal modes analysis.

Nomenclature

Abbreviations

DOF	Degree of freedom
FEM	Finite element model
IRS	Improved reduced system

Matrix Names

G	Transformation
K	Stiffness
M	Mass
P	Applied load
U	Displacement
U^*	Displacement correction
$\tilde{U}$	Approximate displacement
$\hat{U}$	Improved approximate displacement
Ω	Modal frequency
ω	Approximate modal frequency

Set Notation

a	Analysis DOF (f - o)
f	Free (unrestrained) DOF (g - m - s)
g	All structural DOF
m	Dependent DOF from MPC and rigid elements
o	Omitted DOF
s	Restrained DOF from SPC and AUTOSPC

Introduction

Matrix reduction is an established and valuable technique for efficient dynamic analysis of large finite element models. The mass and stiffness matrices of the complete finite element model (FEM) containing thousands of DOF are reduced to a much smaller set (usually hundreds of DOF). This permits much more efficient modal analysis using the Givens or Householder methods. It is also essential for modal survey test/analysis correlation [1] in order to reduce the FEM matrices to DOF corresponding to accelerometer locations in the test.

The standard method for matrix reduction was developed by Guyan [2]. This method is often called "static" reduction since the reduction equations are developed using only the stiffness matrix. The Guyan reduction produces an exact reduction of the stiffness matrix but only an approximate reduction of the mass matrix if any DOF with mass are omitted. The loss of mass accuracy using the Guyan reduction method can often be unacceptably large for complex models with distributed mass.

The Improved Reduced System (IRS) method is a new reduction procedure developed by O'Callahan [3]. The IRS method is an extension of the Guyan method and explicitly includes mass effects in the development of the matrix reduction transformation matrix. The addition of mass effects can significantly increase the accuracy of the mass matrix reduction and the computed frequencies and mode shapes.

This paper describes the implementation of the IRS dynamic reduction method in MSC/NASTRAN. The procedure is implemented as a rigid format alter to a normal modes analysis. The mathematical basis for the method is also presented to illustrate the differences between the IRS and Guyan reduction methods.

Overview of Matrix Reduction

Finite element model stiffness and mass matrices are reduced using transformation methods of the form:

$$K_{aa} = G_{fa}^T K_{ff} G_{fa} \tag{1}$$

$$M_{aa} = G_{fa}^T M_{ff} G_{fa} \quad (2)$$

The transformation matrix G_{fa} is the key of the matrix reduction. It relates the motion of the f-set DOF to the motion of the a-set DOF:

$$U_f = G_{fa} U_a \quad (3)$$

The transformation matrix used by the Guyan reduction can be developed based on a static solution:

$$P_f = K_{ff} U_f \quad (4)$$

Partition (4) into a-set and o-set partitions:

$$\begin{bmatrix} P_o \\ P_a \end{bmatrix} = \begin{bmatrix} K_{oo} & K_{oa} \\ K_{ao} & \overline{K_{aa}} \end{bmatrix} \begin{bmatrix} U_o \\ U_a \end{bmatrix} \quad (5)$$

The overbar for K_{aa} in (5) is to distinguish that this matrix is the a-set partition of K_{ff} rather than the reduced a-set stiffness matrix in (1). Similar terminology will be used later for the mass matrix.

The upper partition of (5) can be written as:

$$P_o = K_{oo} U_o + K_{oa} U_a \quad (6)$$

(6) can be solved to determine the displacements of the o-set DOF in terms of the displacements of the a-set DOF and the o-set forces:

$$U_o = -K_{oo}^{-1} K_{oa} U_a + K_{oo}^{-1} P_o \quad (7)$$

Assuming that there are no forces on the o-set DOF, (7) can be simplified as:

$$U_o = -K_{oo}^{-1} K_{oa} U_a \quad (8)$$

The relationship between the f-set and a-set DOF can be written as

$$U_f = \begin{bmatrix} U_o \\ U_a \end{bmatrix} \quad (9)$$

or using (8):

$$U_f = \begin{bmatrix} -K_{oo}^{-1} K_{oa} \\ I \end{bmatrix} U_a \quad (10)$$

Therefore, the transformation matrix in (3) used by the Guyan reduction is:

$$G_{fa_{Guyan}} = \begin{bmatrix} -K_{oo}^{-1} K_{oa} \\ I \end{bmatrix} \quad (11)$$

The mass and stiffness matrices can now be reduced using the Guyan transformation matrix (11) and the transformation equations (1) and (2). For efficiency, (1), (2), and (11) can be rewritten as derived in [2] and [4]:

$$K_{aa} = \overline{K_{aa}} + K_{oa}^T G_{oa} \quad (12)$$

$$M_{aa} = \overline{M_{aa}} + M_{oa}^T G_{oa} + G_{oa}^T M_{oa} + G_{oa}^T M_{oo} G_{oa} \quad (13)$$

where

$$G_{oa} = -K_{oo}^{-1} K_{oa} \quad (14)$$

The main limitation of the Guyan reduction method is the assumption that there are no forces on the o-set DOF. If there are no o-set forces, the Guyan reduction is exact. However, if there are o-set forces, the Guyan reduction is only approximate with the degree of error based on the magnitude of the o-set forces. The approximation can be exactly corrected in a static analysis by including the second term in (7):

$$U_o^* = K_{oo}^{-1} P_o \quad (15)$$

However, this correction cannot be applied directly to a normal modes analysis since the o-set forces represent inertia terms of the modes which have not yet been calculated. Therefore, if any of the o-set DOF have mass or inertia, the Guyan reduction for normal modes analysis will be approximate.

The Improved Reduced System reduction procedure developed by O'Callahan [3] is an extension to the Guyan method. The IRS method includes a correction term for modal analysis similar to the correction term for a static analysis in (15).

The basis for the IRS reduction method is the standard eigenvalue equation:

$$K_{ff} U_f = M_{ff} U_f \Omega^2 \quad (16)$$

Using the Guyan transformation (11), (16) can be reduced to the a-set DOF and solved:

$$K_{aa} \tilde{U}_a = M_{aa} \tilde{U}_a \omega^2 \quad (17)$$

Note that, due to the mass matrix approximation of the Guyan reduction, the mode shapes and frequencies from (16) and (17) will not be identical. The degree of difference between the two solutions will depend on the a-set DOF.

The f-set approximate mode shapes can be recovered using (17), (11), and (3):

$$\widetilde{U}_f = G_{fa_{Guyan}} \widetilde{U}_a \tag{18}$$

The key feature of the IRS method is the development of a correction term for modal analysis similar to the correction term used by the Guyan reduction for static analysis. An "inertia" force term can be formed by combining (4), (17), and (18):

$$K_{ff} \widetilde{U}_f = \widetilde{P}_f = M_{ff} \widetilde{U}_f \, \omega^2 \tag{19}$$

The dynamic correction term similar to the static correction term can be written using (15) and (19):

$$U_f^* = K_{ff}^{-1} M_{ff} G_{fa_{Guyan}} \widetilde{U}_a \, \omega^2 \tag{20}$$

The improved mode shapes can be written using (18) and (20):

$$\widehat{U}_f = G_{fa_{Guyan}} \widetilde{U}_a + K_{ff}^{-1} M_{ff} G_{fa_{Guyan}} \widetilde{U}_a \, \omega^2 \tag{21}$$

(17) can be rewritten as:

$$M_{aa}^{-1} K_{aa} \widetilde{U}_a = \widetilde{U}_a \, \omega^2 \tag{22}$$

Substitute (22) into (21):

$$\widehat{U}_f = G_{fa_{Guyan}} \widetilde{U}_a + K_{ff}^{-1} M_{ff} G_{fa_{Guyan}} M_{aa}^{-1} K_{aa} \widetilde{U}_a \tag{23}$$

Rearranging terms in (23), the transformation matrix used by the IRS dynamic reduction method can be written as:

$$G_{fa_{IRS}} = \begin{bmatrix} G_{Static} + G_{Dynamic} \\ I \end{bmatrix} \tag{24}$$

where

$$G_{Static} = -K_{oo}^{-1} K_{oa} \tag{25}$$

$$G_{Dynamic} = K_{oo}^{-1} \left[M_{oa} + M_{oo} G_{Static} \right] M_{aa}^{-1} K_{aa} \tag{26}$$

The mass and stiffness matrices can now be reduced using the IRS transformation matrix (24) and the transformation equations (1) and (2).

The key to whether the IRS method will result in a more accurate reduction depends on the dynamic correction terms in (26). The M_{aa} matrix must be invertable and well conditioned. In addition, the assumption that the inertia terms can be calculated based on the reduced a-set problem (17) may not work well for all models. Therefore, the IRS reduction procedure will result in more accurate reduction than the Guyan method if the above conditions are satisfied. However, the IRS method might result in worse answers for certain unique problems.

Implementation

The IRS reduction method is implemented in MSC/NASTRAN as a rigid format alter to a normal modes analysis. The procedure works as follows:

- Perform a standard Guyan (static) reduction of the stiffness and mass matrix (SEKR and SEMR respectively).
- Calculate the IRS transformation matrix using (24).
- Calculate new K_{aa} and M_{aa} using (1), (2), and (24).
- Store the new K_{aa}, M_{aa}, and G_{oa} into the data base.

The remainder of the normal modes analysis is performed using the standard capabilities of the rigid format.

The rigid format alter to implement the IRS dynamic reduction method is included in the appendix of this paper. The alter works with SOL 63 in MSC/NASTRAN Version 65. Similar alters can be developed for other rigid formats or MSC/NASTRAN Version 66.

Example Problems

Two example problems were analyzed to demonstrate the effectiveness of the IRS reduction method. The first problem was a simple beam as shown in Figure 1. This model was similar to the problem analyzed by O'Callahan in [3]. The problem was analyzed using an a-set which was intentionally poor to exaggerate the effectiveness of the IRS correction term. The model was analyzed using no reduction (Lanczos), Guyan reduction, and IRS reduction.

As shown in Table 1, the IRS method provided much superior answers compared to the Guyan reduction. Even using an intentionally poor a-set, the IRS method was able to provide surprisingly accurate answers for the simple problem. The cross-orthogonality between the unreduced and reduced models was also substantially improved by the IRS reduction method as shown in Tables 2 and 3.

The second problem was a generic solid rocket motor shown in Figure 2. A small but reasonable a-set was selected which included 32 of the 1654 f-set DOF. These locations were similar to where accelerometers would be installed in a modal survey test. All accelerometers were placed on the outside casing and nozzle. No a-set points were selected on the interior of the propellant since no accelerometers could be easily installed at these locations. The problem was analyzed using no reduction (Lanczos), Guyan reduction, and IRS reduction.

The IRS reduction method again provided more accurate answers than the Guyan reduction. As shown in Table 4, the frequency errors for the IRS method were significantly smaller than those of the Guyan reduction method. The cross-orthogonality values were also significantly improved for most modes as shown in Tables 5 and 6. However, the IRS method could not adequately correct the final two modes since there were no a-set DOF on the propellant which dominated these modes.

The two example models both showed improved results using the IRS method compared to the Guyan reduction. However, the IRS method will not necessarily always provide good answers. An initially poor a-set selection will limit the correction accuracy. In addition, there may be certain types of models which do not satisfy the assumptions of the dynamic correction term used by the IRS method (see equations 19 through 23). The IRS method should be used with care to verify that it is applicable to specific models.

Conclusions

The IRS reduction method has been implemented for MSC/NASTRAN. The implementation uses a rigid format alter to normal modes analysis.

The IRS method will provide more accurate answers than the Guyan reduction method for many problems. However, because of the assumptions involved in adding mass effects to the matrix reduction transformation matrix, the IRS reduction method may not provide superior results for all models. The method should be used with care and the results closely checked to ensure satisfactory performance.

References

1) Kammer, D.C., Flanigan, C.C., and Dreyer, W., "A Superelement Approach to Test-Analysis Model Development," Proceedings of the 4th International Modal Analysis Conference, Los Angeles, California, February, 1986.

2) Guyan, R.J., "Reduction of Stiffness and Mass Matrices," AIAA Journal, Vol. 3, No. 2, February, 1965.

3) O'Callahan, J., "A Procedure for an Improved Reduced System (IRS) Model," Proceedings of the 7th International Modal Analysis Conference, Las Vegas, Nevada, February, 1989.

4) The NASTRAN Theoretical Manual, R. H. MacNeal, editor, The MacNeal-Schwendler Corporation, 1972, Section 3.5.



Figure 1. The first example problem was a uniform beam with an intentionally poor selection of a-set DOF.

Table 1. Frequency agreement for the uniform beam example problem.

Mode Number	Exact (Lanczos)	Guyan Reduction	IRS Reduction	Mode Shape Description
1	8.992	9.000	8.992	1st bending
2	55.724	57.709	55.728	2nd bending
3	154.466	205.958	159.031	3rd bending

Table 2. Cross-orthogonality of the Guyan reduction results.

		Exact		
		1	2	3
Guyan	1	100	-6	44
	2	0	-93	-49
	3	0	0	-49

Note: Cross-orthogonality = $\phi_{aGuyan}{}^{T} M_{aaGuyan} \phi_{aExact}$

Table 3. Cross-orthogonality of the IRS reduction results.

		Exact		
		1	2	3
IRS	1	-100	0	-22
	2	0	100	23
	3	0	0	76

Note: Cross-orthogonality = $\phi_{aIRS}{}^{T} M_{aaIRS} \phi_{aExact}$



Figure 2. The second example problem was a generic solid rocket motor.

Table 4. Frequency agreement for the solid rocket motor example problem

Mode Number	Exact (Lanczos)	Guyan Reduction	IRS Reduction	Mode Shape Description
1	0.0	0.0	0.0	Rigid body mode
2	0.0	0.0	0.0	Rigid body mode
3	0.0	0.0	0.0	Rigid body mode
4	63.464	65.758	63.464	1st nozzle ovaling (N=2)
5	69.523	71.174	69.531	1st bending
6	79.741	87.939	79.747	1st case ovaling (N=2, M=1)
7	93.118	104.318	93.142	2nd case ovaling (N=2, M=2)
8	103.248	108.888	103.254	2nd nozzle ovaling (N=3)
9	132.404	153.407	132.496	3rd case ovaling (N=2, M=3)
10	134.752	not found	142.750	1st propellant / case axial
11	149.005	199.483	158.043	2nd bending / propellant shear

Table 5. Cross-orthogonality of the Guyan reduction results.

			E	X	A	C	T		
		4	5	6	7	8	9	10	11
	4	-92	0	1	0	0	1	0	0
G	5	0	95	0	0	3	0	0	2
U	6	0	0	82	12	0	-8	0	0
Y	7	0	0	-5	79	0	8	0	0
A	8	0	0	0	0	-88	0	0	4
N	9	0	0	0	1	0	-73	0	0
	10								
	11	0	0	0	0	0	0	0	-41

Note: Cross-orthogonality = $\phi_{aGuyan}^{T} M_{aaGuyan} \phi_{aExact}$

Table 6. Cross-orthogonality of the IRS reduction results.

			E	X	A	C	T		
		4	5	6	7	8	9	10	11
	4	-99	0	0	0	0	0	0	0
	5	0	100	0	0	0	0	0	1
I	6	0	0	100	0	0	0	0	1
R	7	0	0	0	99	0	1	0	0
S	8	0	0	0	0	-99	0	0	0
	9	0	0	0	0	0	99	0	0
	10	0	0	0	0	0	0	50	0
	11	0	0	0	0	0	0	0	-64

Note: Cross-orthogonality = $\phi_{aIRS}^{T} M_{aaIRS} \phi_{aExact}$

APPENDIX

Rigid Format Alter for the IRS Dynamic Reduction Method

```
$  IRSRED - REDUCE MASS AND STIFFNESS MATRICES USING IRS METHOD
$  ------------------------------------------------------------
$
$  THIS ALTER REDUCES THE STIFFNESS AND MASS MATRICES OF THE
$  RESIDUAL STRUCTURE USING THE "IMPROVED REDUCED SYSTEM" METHOD.
$  THIS METHOD IMPROVES UPON THE STANDARD GUYAN (STATIC)
$  REDUCTION BY INCLUDING THE EFFECTS OF MASS AT THE O-SET DOF.
$
$
$  REFERENCE:
$
$    O'CALLAHAN, J., "A PROCEDURE FOR AN IMPROVED REDUCED SYSTEM
$    (IRS) MODEL," 7TH INTERNATIONAL MODAL ANALYSIS CONFERENCE,
$    FEBRUARY, 1989.
$
$
$  REQUIREMENTS TO USE THIS ALTER -
$-----------------------------------------------------------------------
$  DATA BASES -
$    NO SPECIAL REQUIREMENTS
$-----------------------------------------------------------------------
$  EXECUTIVE DECK -
$    SOL 63
$    THIS ALTER
$    DIAG 8,20 RECOMMENDED
$-----------------------------------------------------------------------
$  CASE CONTROL DECK -
$    INCLUDE SEMG, SEKR, AND SEMR (OR SEALL) REQUESTS FOR THE
$    THE RESIDUAL STRUCTURE.
$
$    THIS ALTER SHOULD *NOT* BE INCLUDED WHEN PROCESSING UPSTREAM
$    COMPONENTS.
$-----------------------------------------------------------------------
$  BULK DATA DECK -
$
$    INCLUDE ASET OR OMIT CARDS TO SELECT THE A-SET DOF.
$
$    DO NOT INCLUDE Q-SET DOF FOR THE RESIDUAL STRUCTURE.
$-----------------------------------------------------------------------
$  EXAMPLE NASTRAN DECK-
$
$    ID      TAM,IRSRED
$    SOL     63
$    TIME    30
$    DIAG    8,20
$    .
$    .  THIS ALTER
$    .
$    CEND
$    TITLE    = REDUCE RESIDUAL STRUCTURE USING IRS METHOD
$    $
$    SET 1000 = 0               $ RESIDUAL STRUCTURE
$    SEALL    = 1000            $ ALL S.E. OPS FOR S.E. IN SET 1000
$    $
$    $
$    SUBCASE 1100
$      SUPER 100
$      LABEL = ANTENNA
$      METHOD = 50
$    SUBCASE 1200
```

(page 1 of 3)

IRS dynamic reduction alter for SOL 63, MSC/NASTRAN Version 65.

```
$       SUPER 200
$       LABEL = BUS STRUCTURE
$       METHOD = 50
$     SUBCASE 2000
$       LABEL = RESIDUAL STRUCTURE
$       METHOD = 25
$     $
$     BEGIN BULK
$     .
$     .  TAM BULK DATA INCLUDING ASET CARDS
$     .
$     ENDDATA
$
$-----------------------------------------------------------------------
$  HISTORY DOCUMENTATION -
$
$     VERSION 1.0   04-APR-89   CHRIS FLANIGAN
$       - ORIGINAL VERSION
$
$=======================================================================
$234567890123456789012345678901234567890123456789012345678901234567890123456789012
$        1         2         3         4         5         6         7
$
$  DEFAULT VALUES FOR PARAMETERS
$
ALTER    1 $                                   AFTER BEGIN
PARAM    //C,N,NOP/V,Y,NOIRSRED=0 $            0 = USE IRS METHOD
$
$  STORE KFF FOR LATER USE
$
ALTER    381 $ V65                             AFTER LBL3
DBSTORE  KFF//MODEL/SEID/DBSET1 $              STORE KFF
$
$  CHANGE THE NAME OF GOAT AND KAA TO GOATS AND KAAS
$  (STATIC CONTRIBUTION)
$
ALTER    420,421 $ V65                         REPLACE FBS AND MPYAD
FBS      LOO,UOO,KOA/GOATS/-1/-1/0/0 $         FORM GOAT (STATIC)
MPYAD    KAO,GOATS,KAA1/KAAS/////UNSYM $       KAA (STATIC)
ALTER    424,424 $ V65                         REPLACE DBSTORE
DBSTORE  GOATS//MODEL/SEID/DBSET1 $            STORE GOAT (STATIC)
ALTER    426,426 $ V65                         REPLACE EQUIV
EQUIV    KAAS,KTT/NOQSET $                     EQUIV IF NO Q-SET
ALTER    429,429 $ V65                         REPLACE UPARTN
UPARTN   USET,KAAS/KQQ1,,,KTT/C,N,A/C,N,Q/C,N,T $ EXTRACT KQQ
ALTER    436,436 $ V65                         REPLACE DBSTORE
DBSTORE  KAAS//MODEL/SEID/DBSET2 $             STORE KAA (STATIC)
$
$  USE GOATS INSTEAD OF GOAT TO PERFORM THE STATIC
$  REDUCTION OF THE MASS MATRIX
$
ALTER    488,488 $ V65                         REPLACE DBSTORE
DBFETCH  /GM,GOATS,DM,USET,/MODEL/PEID//DBSET3 $ FETCH DATA
ALTER    490,490 $ V65                         REPLACE EQUIV
EQUIV    GOATS,GOA/NP $                        EQUIV
ALTER    511,511 $ V65                         REPLACE MPYAD
MPYAD    MOO,GOATS,MOA/MOA1 $                  REDUCE O-SET MASS
ALTER    516,516 $ V65                         REPLACE EQUIV
EQUIV    MTT,MAAS/NOQSET $                     EQUIV IF NO Q-SET
ALTER    518,518 $ V65                         REPLACE UMERGE1
```

(page 2 of 3)

IRS dynamic reduction alter for SOL 63, MSC/NASTRAN Version 65.

```
UMERGE1  USET,MTT,,,/MAAS/C,N,A/C,N,T/C,N,Q/0 $  SYMMETRIC MERGE
ALTER    521,522 $ V65                            REPLACE MPYAD
MPYAD    MOA,GOATS,MAA1/MAA2/1 $                  REDUCE O-SET MASS
MPYAD    GOATS,MOA1,MAA2/MAAS/1////6 $            REDUCE O-SET MASS
ALTER    524,524 $ V65                            REPLACE EQUIV
EQUIV    MAAS,MTT/NOQSET $                        EQUIV IF NO Q-SET
ALTER    527,527 $ V65                            REPLACE UPARTN
UPARTN   USET,MAAS/MTT,,,MQQ1/C,N,A/C,N,T/C,N,Q $ SYMMETRIC PARTN
ALTER    535,535 $ V65                            REPLACE DBSTORE
DBSTORE  MAAS//SOLID/SEID/DBSET2 $                STORE MAA (STATIC)
$
$  CALCULATE THE IMPROVED TRANSFORMATION MATRIX
$
$    GOAT = GOATS + GOATD
$
$                 -1
$    GOATS = -KOO    KOA
$
$                -1                          -1
$    GOATD = KOO    (MOO * GOATS + MOA) MAA    KAA
$
$  NOTE:  THE A-SET MASS MATRIX *MUST* BE POSITIVE DEFINITE
$         SUCH THAT MAA CAN BE INVERTED.  ALL ZERO MASS DOF
$         *MUST* BE OMITTED.
$
DBFETCH  /KFF,LOO,UOO,GOATS,KAAS/MODEL/PEID//DBSET3 $ FETCH SEKR DATA
SOLVE    MAAS,KAAS/MINVK $                        MAAS-INV * KAAS
MPYAD    MOO,GOATS,MOA/MOA1 $                     (MOO*GOATS) + MOA
MPYAD    MOA1,MINVK,/MOA2 $                       MOA1 * (MAAS-INV*KAAS)
FBS      LOO,UOO,MOA2/GOATD $                     O-A TRANSFORM (DYNAMIC)
ADD      GOATS,GOATD/GOAT $                       TOTAL O-A TRANSFORM
DBSTORE  GOAT//MODEL/SEID/DBSET1 $                STORE GOAT
$
$  FORM THE IMPROVED REDUCED STIFFNESS AND MASS MATRICES
$
MATGEN   ,/IAA/1/NOASET $                         A-SET IDENTITY MATRIX
VEC      USET/VFOA/F/O/A $                        F = O / A
MERGE    GOAT,IAA,,,,VFOA/TFA/1 $                 ROW MERGE
SMPYAD   TFA,KFF,TFA,,,/KAA/3////1////6 $         REDUCED K MATRIX
DBSTORE  KAA//MODEL/SEID/DBSET2 $                 STORE KAA
SMPYAD   TFA,MFF,TFA,,,/MAA/3////1////6 $         REDUCED M MATRIX
DBSTORE  MAA//SOLID/SEID/DBSET2 $                 STORE MAA
$
$--END OF IRSRED---------------------------------------------------------
```

(page 3 of 3)

IRS dynamic reduction alter for SOL 63, MSC/NASTRAN Version 65.